
Supplementary material

      SUBROUTINE EALGN(C,EG,AG,E1,A1,E2,A2,E3,A3,E4,A4)
C -------------------------------------------------------------------
C Provides Em,Am (m=g,1,2,...) for the analytical approximation of
C FI-BAR(x;c) - the size profile for spherical crystallites with
C lognormal distribution - different in 4 intervals of c: 
C [0,c12], (c12,c23], (c23,c34] and (c34,6] 
      data c12,c23,c34 /0.372719, 0.780008, 2.03119/
c -------------------------------------------------------------------
C In [0,6]:
      fi1(x)=0.271972+0.112729e-1*x+2.16372*x*exp(-2.32963*x)
     &                             +0.669352*x*exp(-0.700742*x)
c In [c12,c34]:
      fig(x)=-0.206365+1.10723*exp(-x/1.31571)
c In [c12,6]:
      fi2(x)=2.27373
c In [c23,c34], set x=c34-c when calling:
      zetg(x)=0.337662*x-0.348946*x**2+0.146579*x**3
c In [c23,6]:
      fi3(x)=(7.33377-0.894178*x)/(1.-0.167151*x+0.913176e-3*x**3)
c In [c34,6]:
      zet4(x)=0.273193e-4*(exp(x/2.34023)-exp(c34/2.34023))
      fi4(x)=54.0539
c -------------------------------------------------------------------
      w=1+c
      c1=0.5625*w
      c2=1.5/w**2
c -------------------------------------------------------------------
      if(c.gt.c12)go to 2
c ************************* Range 1 - [0,c12]: G+L1 *****************
      f1=fi1(c)
      z1=1.
      w=1.-f1
      fg=(c1-f1**2)/w
      zg=w/fg
      eg=fg*zg
      ag=c2/fg
      e1=f1*z1
      a1=c2/f1
      return
2     if(c.gt.c23)go to 3
c ************************* Range 2 - (c12,c23]: G+L1+L2 *********** 
      fg=fig(c)
      f1=fi1(c)
      f2=fi2(c)
      w=f1**2
      z2=(c1-w-fg*(1.-f1))/(f2-f1)/(f1+f2-fg)
      z1=1.-z2
      zg=(c1-w*z1-f2**2*z2)/fg**2
      eg=fg*zg
      ag=c2/fg
      e1=f1*z1
      a1=c2/f1
      e2=f2*z2
      a2=c2/f2
      return
3     if(c.gt.c34)go to 4
c ************************** Range 3 - (c23,c34]: G+L1+L2+L3 ******* 
      fg=fig(c)



      w=c34-c
      zg=zetg(w)
      f1=fi1(c)
      f2=fi2(c)
      f3=fi3(c)
      w=1.-fg*zg
      ww=fg**2*zg
      z2=(c1+f1*f3-(f1+f3)*w-ww)/(f2-f1)/(f2-f3)
      z3=(c1+f1*f2-(f1+f2)*w-ww)/(f3-f1)/(f3-f2)
      z1=1.-z2-z3
      eg=fg*zg
      ag=c2/fg
      e1=f1*z1
      a1=c2/f1
      e2=f2*z2
      a2=c2/f2
      e3=f3*z3
      a3=c2/f3
      return
4     continue
c ************************** Range 4 - (c34,6}: L1+L2+L3+L4 ******** 
      f1=fi1(c)
      f2=fi2(c)
      f3=fi3(c)
      f4=fi4(c)
      z4=zet4(c)
      w=1.+(f1-f4)*z4
      ww=(f1**2-f4**2)*z4
      z2=(c1+f1*f3-(f1+f3)*w+ww)/(f2-f1)/(f2-f3)
      z3=(c1+f1*f2-(f1+f2)*w+ww)/(f3-f1)/(f3-f2)
      z1=1.-z2-z3-z4
      e1=f1*z1
      a1=c2/f1
      e2=f2*z2
      a2=c2/f2
      e3=f3*z3
      a3=c2/f3
      e4=f4*z4
      a4=c2/f4
      return
      end


